Chapter 12

Contents

Chapters

12.1 Concepts of ADC Interfacing with Sensors

One of the mixed signal features of PIC18 devices is the ADC. Microprocessors and microcontrollers are digital devices and operate on digital values - 1’s and 0’s. The signals from sensors typically are analog in nature and must be converted to digital first, stored into memory and later processed by microprocessors or microcontrollers. This is shown in fig 12.1a.

The signal from sensor is either a constant value or a value that is varying at a specific frequency any given time. The analog input signal is sampled through a sample and hold circuit as shown in fig 12.1b. The switch when closed charges the holding capacitor and holds the value equivalent to analog input even the switch is opened. The sampling frequency is the rate at which this switch is operated. The signal could be captured accurately, only when the sampling frequency is greater or equal to twice of the analog input frequency. Charging time is the time during capacitor charges from VSS to a value equivalent to analog input voltage. This is called as acquisition time. Hold time is the time during which the capacitor should hold the value i.e. time during which the complete A/D conversion is to be performed. This is the minimum time during which one sample could be taken. So according to the sampling theory, this implies that time-period of the analog input signal should be at least twice the total of acquisition time and conversion time without which the analog signal could not be accurately captured.

Here a set of analog values are mapped to a single digital value, and this is called quantization. When the set of analog values is small (1 in ideal ADC), the accuracy is more, otherwise it is less accurate (in practical ADC). When a greater number of binary bits are used to represent an analog value, the set becomes less and the reading is more accurate. A greater number of bits results in higher processing and delay, hence, the number of bits used to represent an analog value is finite. So, a finite number of digital values exists for a range of analog values. These digital values are the steps that may be produced by ADC while converting the analog value into a digital value. A step representation for different analog input voltages in presented in fig 12.2.



Fig 12.1a) Sensor Interfacing to Microcontroller



Fig 12.1b) Sample & Hold Circuit

The process of A/D conversion requires operating voltage V_{CC}, Ground or V_{SS}. Additionally VREF+ & VREF- may be used as reference voltages based on the applied analog voltage. When VREF+ is 5V and VREF- is GND the effective voltage difference is 5V and when VREF+ is 5V and VREF- is 1.7V the effective voltage difference is 3.3V.



Fig 12.2 Sample A/D conversion steps for different analog input voltages

Source: TI ADC 0808

Table 12.1: Resolution values for different voltages and Number of bits for representation.

No of Bits	Steps	Resolution			
		5V	3.3V	1.8V	1.2V
5	$2^5 = 32$	0.1562	0.1031	56.25 mV	37.5 mV
8	$2^8 = 256$	19.53 mV	12.89 mV	7.031 mV	4.687 mV
10	$2^{10} = 1024$	4.882 mV	3.222 mV	1.757 mV	1.171 mV
12	$2^{12} = 4096$	1.22 mV	0.8057 mV	0.4395 mV	0.293 mV
16	$2^{16} = 65536$	76.3 µV	50.4 µV	27.5 µV	18.3 µV

Resolution:

Resolution is the minimum change in input that can produce a detectable change in output from a non-zero input. In A/D conversion the resolution of ADC is defined as the ratio of the effective voltage difference to the maximum number of possible steps. If N is the number bits, the maximum number of possible steps is given by 2 power N i.e., 2^N.

- For a 10-bit ADC with 5V as VREF+ & Ground as VREF-, the resolution is given by
  $5V/(2^{10})$ = 5/1024 = 0.004882813 = 4.883 mV.
- For a 10-bit ADC with 3.3V supply, the resolution is given by
  $3.3V/(2^{10})$ = 3.3/1024 = 0.003222656 = 3.22 mV.

- For a 10-bit ADC with 1.8V supply, the resolution is given by
  1.8V/(2^{10}) = 1.8/1024 = 0.001757813 = 1.758 mV.
- For a 10-bit ADC with 1.2V supply, the resolution is given by
  1.2V/(2^{10}) = 1.2/1024 = 0.001171875 = 1.172 mV.

Here the values presented are ideal. i.e. considers that ADC is ideal and a 4.883mv generates a 10-bit value of 00 0000 0001. Similarly, for 19.53mV the 10-bit converted value is 00 0000 0100. Due to impedance mismatch, parasitic, interconnects, voltage drop etc. voltage presented at the input of ADC may be different and due to which the actual converted value may deviate from ideal values.

The resolution values for different voltages & number of bits for representing Analog Voltage is presented in table 12.1. The different steps, equivalent A/D Binary Value, for a specific Analog Voltage for different voltage ranges is presented Table 12.2.

Table 12.2: Steps, equivalent A/D Binary Value, for a specific input Voltage in different voltage ranges.

Steps	Binary Value	Analog Voltage for respective power supplies			
		5V	3.3V	1.8V	1.2V
0	00 0000 0000	0	0	0	0
1	00 0000 0001	4.882 mV	3.222 mV	1.757 mV	1.171 mV
4	00 0000 0100	19.53 mV	12.89 mV	7.031 mV	4.687 mV
16	00 0001 0000	78.12 mV	51.56 mV	28.12 mV	18.75 mV
32	00 0010 0000	0.1562	0.1031	56.25 mV	37.5 mV
64	00 0100 0000	0.3125	0.2062	0.1135	750 mV
128	00 1000 0000	0.625	0.4125	0.225	0.15
192	00 1100 0000	0.9375	0.6187	0.3375	0.225
256	01 0000 0000	1.25	0.825	0.45	0.3
512	10 0000 0000	2.5	1.65	0.9	0.6
676	10 1010 0100	3.3	2.1785	1.1882	0.792
1000	11 1110 1000	4.882	3.222	1.7578	1.171
1023	11 1111 1111	4.995	3.296	1.7982	1.198
1024	1 00 0000 0000	5	3.3	1.8	1.2

A typical block diagram of an ADC0808 is presented in fig 12.3. A/D conversion is started only after the acquisition of the signal. A typical application has multiple sensor signals which are required to be converted to digital values. Hence ADCs are provided with capability to connect multiple inputs (8) (VIN1 – VIN8 or In0 – In7) and the address select lines (3) (A, B & C) or (AD0-AD2) are provided to select one of the inputs for acquisition and conversion. The ALE signal acts as strobe pulse for channel selection and selects a specific channel based on the address select lines.



Fig 12.3 Block diagram of ADC 0808

Source: TI ADC 0808

The ADC operates on the clock frequency and to limit the conversion time a higher clock frequency is required. As ADC's conversion depends on the number of bits and the response of inbuilt DAC and comparators, the operating frequency is to be within in a specified range. The timing different signals for A/D conversion is shown in fig 12.4.

A "start of conversion" – SOC (GO bit PIC18) signal initiates the A/D conversion process. The "end of conversion" – EOC (DONE bit in PIC18) signal indicates that the conversion process is complete and could be used to generate an interrupt (ADIF in PIC18). The "output enable" OE signal acts a read signal and activates the output buffer to place the ADC's converted value on to the data bus. A typical interfacing of the ADC0808 with a Microprocessor/PPI/Microcontroller is shown in fig 12.5.

This external interfacing is not required when microcontrollers have an in-built ADC. The interfacing pins may be used as I/O's for connecting any other external devices. Also, the equivalent ADC signal pins become control/status bits in registers which are used to control the ADC. This is discussed in the following PIC18 ADC section.



Fig 12.4 Waveform of a A/D conversion using ADC 0808

Source: TI ADC 0808



Fig 12.5: Interfacing of ADC0808 with Microprocessor/PPI/Microcontroller

12.2 PIC18's 10-Bit ADC & Configuration

Classic microprocessors/microcontrollers do not have an in-built ADC, require an external ADC interface whereas PIC18 devices have one in-built ADC that can convert an analog signal into a 10-bit digital value. PIC devices presented here have 10-bit SAR type ADC with input channels varying between 8 and 16 as shown below in table 12.3.

Table 12.3: ADC Registers & Channels in PIC Devices

Device	No of I/P Channels	Resolution	Registers	
PIC18F458	**8**	10-Bit	ADCON0 ADCON1	ADRESH ADRESL
PIC18C801	**12**	10-Bit	ADCON0 ADCON1 **ADCON2**	
PIC18F8720	**16**	10-Bit	ADCON0 ADCON1 **ADCON2**	

ADCON0: A/D Control Register 0 Format (PIC18F458)

R/W-0	R/W-0	R/W-0	R/W-0	R/W-0	R/W-0	U-0	R/W-0
ADCS1	ADCS0	CHS2	CHS1	CHS0	GO/$\overline{\text{DONE}}$	—	ADON
bit 7							bit 0

Bit No	Bit	Description
bit 7-6	ADCS1:ADCS0	**A/D Conversion Clock Select bits:** Refer to Table 12.5
bit 5-3	CHS2:CHS0	**Analog Channel Select bits:** Refer to Table 12.4
bit 2	GO/$\overline{\text{DONE}}$	**A/D Conversion Status bit** When ADON = 1: 1 = A/D conversion in progress (setting this bit starts the A/D conversion which is automatically cleared by hardware when the A/D conversion is complete) 0 = A/D conversion not in progress
bit 1	Unimplemented	Read as '0'
bit 0	ADON	**A/D ON bit** 1 = A/D converter module is powered up 0 = A/D converter module is shut-off and consumes no operating current

ADCON1: A/D Control Register 1 Format (PIC18F458)

R/W-0	R/W-0	U-0	U-0	R/W-0	R/W-0	R/W-0	R/W-0
ADFM	ADCS2	—	—	PCFG3	PCFG2	PCFG1	PCFG0
bit 7							bit 0

Bit No	Bit	Description

bit 7	ADFM	**A/D Result Format Select bit** 1 = Right justified. Six (6) Most Significant bits of ADRESH is read as ‘0’. 0 = Left justified. Six (6) Least Significant bits of ADRESL are read as ‘0’.
bit 6	ADCS2	**A/D Conversion Clock Select bit**: Refer to table 12.5. This bit is linked with bit 7-6 ADCS1:ADCS0 of ADCON0
bit 5-4	Unimplemented	Read as ‘0’
bit 3-0	PCFG3:PCFG0	**A/D Port Configuration Control bits:** Refer to table 12.6

The A/D converter modules of PIC18F458, PIC18C801 & PIC18F8720 have similar operation & construction except for the differences in number of input channels & control registers for configuring it. The ADC variations of different devices are summarised and presented in table 12.3. The channel-select bits and corresponding channel selection for different devices is presented in Table 12.4. The registers associated with ADC module are

- A/D Control Register 0 (ADCON0)
- A/D Control Register 1 (ADCON1)
- A/D Control Register 2 (ADCON2) → In PIC18C801 & PIC18F8720 only
- A/D Result High (ADRESH) & A/D Result Low (ADRESL) Registers

Table 12.4 Analog Channel Select bits CHS3:CHS0

Ch. No	AN No	PIC18F458		PIC18C801		PIC18F8720	
		CHS2:CHS0	Port Pin	CHS3:CHS0	Port Pin	CHS3:CHS0	Port Pin
1	AN0	000	RA0	0000	RA0	0000	RA0
2	AN1	001	RA1	0001	RA1	0001	RA1
3	AN2	010	RA2	0010	RA2	0010	RA2
4	AN3	011	RA3	0011	RA3	0011	RA3
5	AN4	100	RA5	0100	RA5	0100	RA5
6	AN5	101	RE0	0101	RF0	0101	RF0
7	AN6	110	RE1	0110	RF1	0110	RF1
8	AN7	111	RE2	0111	RF2	0111	RF2
9	AN8			1000	RH4	1000	RF3
10	AN9			1001	RH5	1001	RF4
11	AN10			1010	RH6	1010	RF5
12	AN11			1011	RH7	1011	RF6
13	AN12					1100	RH4
14	AN13					1101	RH5
15	AN14					1110	RH6
16	AN15					1111	RH7

Ch. No implies Channel Number, AN No implies Analog Input Number.

Note: In PIC18F458 CHS2:CHS0 bits are present in register ADCON0<5:3>. In PIC18C801 & PIC18F8720 the CHS3:CHS0 bits are present in register ADCON2<5:2>.

Table 12.5 A/D Conversion Clock Select bits ADCS2:ADCS0

ADCS2 (1)	ADCS1:ADCS0 (1)	Clock Conversion
0	00	FOSC/2
0	01	FOSC/8
0	10	FOSC/32
0	11	FRC: Clock derived from internal A/D RC oscillator 1 MHz max
1	00	FOSC/4
1	01	FOSC/16
1	10	FOSC/64
1	11	FRC: Clock derived from internal A/D RC oscillator 1 MHz max

Note: 1) In PIC18F458 ADCS2 bit is present in register ADCON1<6> and ADCS1:ADCS0 bits are present in register ADCON0<7:6>. In PIC18C801 & PIC18F8720 the ADCS2:0 bits are present in register ADCON2<2:0>.

Table 12.6 A/D Port Configuration Control bits PCFG3:PCFG0 of PIC18F458.

PCFG3: PCFG0	Analog Input - ANx								Power Supply		A/V
	7	6	5	4	3	2	1	0	VREF+	VREF-	
0000	A	A	A	A	A	A	A	A	VDD	VSS	8/0
0001	A	A	A	A	VREF+	A	A	A	AN3	VSS	7/1
0010	D	D	D	A	A	A	A	A	VDD	VSS	5/0
0011	D	D	D	A	VREF+	A	A	A	AN3	VSS	4/1
0100	D	D	D	D	A	D	A	A	VDD	VSS	3/0
0101	D	D	D	D	VREF+	D	A	A	AN3	VSS	2/1
0110	D	D	D	D	D	D	D	D	—	—	0/0
0111	D	D	D	D	D	D	D	D	—	—	0/0
1000	A	A	A	A	VREF+	VREF-	A	A	AN3	AN2	6/2
1001	D	D	A	A	A	A	A	A	VDD	VSS	6/0
1010	D	D	A	A	VREF+	A	A	A	AN3	VSS	5/1
1011	D	D	A	A	VREF+	VREF-	A	A	AN3	AN2	4/2
1100	D	D	D	A	VREF+	VREF-	A	A	AN3	AN2	3/2
1101	D	D	D	D	VREF+	VREF-	A	A	AN3	AN2	2/2
1110	D	D	D	D	D	D	D	A	VDD	VSS	1/0
1111	D	D	D	D	VREF+	VREF-	D	A	AN3	AN2	1/2

A-Analog, D-Digital and A/V – Number of Analog channels/Number of voltage References.

Note: The PCFG3:PCFG0 bits configuration in PIC18F458 varies from that of PIC18C801 & PIC18F8720 even though the bits are in the same register – ADCON1<3:0> in all the devices.

The PIC18F458 uses two A/D control registers ADCON0, ADCON1. The PIC18C801 and PIC18F8720 uses three A/D control registers ADCON0, ADCON1 & ADCON2. The control registers are used to control/configure the input channel, clock selection, reference voltage selection, start of conversion and indicate end of a conversion etc. The A/D converted result in stored in A/D Result High & Low (ADRESH & ADRESL) register pair. The general block diagram of an ADC is shown in Fig 12.6.

The control register formats for configuring the ADC of PIC18F458 is presented earlier. The port configuration control bits configure port pins related to ADC as analog, digital or power supply reference pins. This port configuration-select bits and their respective configuration is shown in Table 12.6.

The formats of PIC18C801 & PIC18F8720 are presented next. The control function of the 3 registers here is similar to that of the 2 registers in PIC18F458. The variation comes due to the re-organization of control bits and the new selection bits due to the additional channels and power supply configurability.

ADCON0: A/D Control Register 0 Format (PIC18C801 & PIC18F8720)

U-0	U-0	R/W-0	R/W-0	R/W-0	R/W-0	R/W-0	R/W-0
—	—	CHS3	CHS2	CHS1	CHS0	GO/$\overline{\text{DONE}}$	ADON
bit 7							bit 0

Bit No	**Bit**	**Description**
bit 7-6	Unimplemented	Read as ‘0’
bit 5-2	CHS3:CHS0	**Analog Channel Select bits.** Refer to Table 12.4
bit 1	GO/DONE	**A/D Conversion Status bit** When ADON = 1: 1 = A/D conversion in progress (setting this bit starts the A/D conversion, which is automatically cleared by hardware when the A/D conversion is complete) 0 = A/D conversion not in progress
bit 0	ADON	**A/D On bit** 1 = A/D converter module is enabled 0 = A/D converter module is disabled

ADCON1: A/D Control Register 1 Format (PIC18C801 & PIC18F8720)

U-0	U-0	R/W-0	R/W-0	R/W-0	R/W-0	R/W-0	R/W-0
—	—	VCFG1	VCFG0	PCFG3	PCFG2	PCFG1	PCFG0
bit 7							bit 0

Bit No	**Bit**	**Description**
bit 7-6	Unimplemented	Read as ‘0’
bit 5-4	VCFG1:VCFG0	**Voltage Reference Configuration bits:**

		Refer to table 12.8
bit 3-0	PCFG3:PCFG0	**A/D Port Configuration Control bits**
		Refer to table 12.7



Fig 12.6 Block Diagram of A/D conversion in PIC Devices.

Operation of ADC

When the PIC device is reset, the A/D module is turned OFF, and any conversion process is aborted. The PIC's SAR (Successive Approximation Register) A/D module internally has a sample/hold circuit, and the output of this sample/Hold circuit is provided as an input to the ADC for conversion.

ADCON2: A/D Control Register 2 Format (PIC18C801 & PIC18F8720)

R/W-0	U-0	U-0	U-0	U-0	R/W-0	R/W-0	R/W-0
ADFM	—	—	—	—	ADCS2	ADCS1	ADCS0
bit 7							bit 0

Bit No	Bit	Description
bit 7	ADFM	**A/D Result Format Select bit** 1 = Right justified 0 = Left justified
bit 6-3	Unimplemented	Read as '0'
bit 2-0	ADCS2:ADCS0	**A/D Conversion Clock Select bits**. Refer to Table 12.5

Table 12.7 A/D Port Configuration Control bits PCFG3:PCFG0 of PIC18C801 & 18F8720.

PCFG3: PCFG0	Analog Input – ANx															
	15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
0000	A	A	A	A	A	A	A	A	A	A	A	A	A	A	A	A
0001	D	D	A	A	A	A	A	A	A	A	A	A	A	A	A	A
0010	D	D	D	A	A	A	A	A	A	A	A	A	A	A	A	A
0011	D	D	D	D	A	A	A	A	A	A	A	A	A	A	A	A
0100	D	D	D	D	D	A	A	A	A	A	A	A	A	A	A	A
0101	D	D	D	D	D	D	A	A	A	A	A	A	A	A	A	A
0110	D	D	D	D	D	D	D	A	A	A	A	A	A	A	A	A
0111	D	D	D	D	D	D	D	D	A	A	A	A	A	A	A	A
1000	D	D	D	D	D	D	D	D	D	A	A	A	A	A	A	A
1001	D	D	D	D	D	D	D	D	D	D	A	A	A	A	A	A
1010	D	D	D	D	D	D	D	D	D	D	D	A	A	A	A	A
1011	D	D	D	D	D	D	D	D	D	D	D	D	A	A	A	A
1100	D	D	D	D	D	D	D	D	D	D	D	D	D	A	A	A
1101	D	D	D	D	D	D	D	D	D	D	D	D	D	D	A	A
1110	D	D	D	D	D	D	D	D	D	D	D	D	D	D	D	A
1111	D	D	D	D	D	D	D	D	D	D	D	D	D	D	D	D

Shaded cells/Inputs are not present in PIC18C801, contains only 0-11 inputs

A-Analog, D-Digital,

Note: The PCFG3:PCFG0 bits configuration in PIC18C801 & PIC18F8720 varies from that of PIC18F458 even though the bits are in the same register – ADCON1<3:0> in all the devices.

A stable reference voltage is required before a conversion is performed. Typically, the first step in the operation of ADC is the selection of analog reference voltage as it is software selectable through the PCFG3: PCFG0 bits of ADCON1<3:0> register of Table 12.6 in PIC18F458 and through the VCFG1:VCFG0 bits of ADCON1<5:4> register i.e. table 12.8 in PIC18C801 & 18F8720. Using these bits, the reference voltage for ADC could be set to device's internal positive (AVDD) / negative (AVSS) supply source or the external voltage source available through the RA3/AN3/VREF+ pin and RA2/AN2/VREF- pins.

Table 12.8 Voltage Reference Configuration bits PCFG3:PCFG0 of PIC18C801 & 18F8720

VCFG1:VCFG0	A/D VREF+	A/D VREF-
00	AVDD	AVSS
01	External VREF+	AVSS
10	AVDD	External VREF-
11	External VREF+	External VREF-

Since the voltage reference may be selected from an external source and the analog signals are applied as inputs, the corresponding port pins must be configured as analog/digital/voltage input pins. This is done through the PCFG3: PCFG0 bits of ADCON1<3:0> register. Table 12.6 provides the configurable settings for PIC18F458, and the table 12.7 provides the options for PIC18C801 & 18F8720. The corresponding analog input channel port pins must be first configured as inputs by clearing the TRIS bits according to table 12.2. i.e. all analog inputs pins must be set as inputs through TRIS bits.

To perform conversion at a specific rate/frequency a clock signal is required. The appropriate clock is selected through the A/D Conversion Clock. Select bits ADCS2:ADCS0 as shown in table 12.3. After the selection of the clock the ADC module may be turned ON for operation by setting the ADON bit in ADCON<0> register.

Since the SAR ADC module acquires signal from Sample/Hold circuit, and operates on the selected clock, a minimum acquisition time is required before a conversion is started. This is known as acquisition time. The ADIF flag in PIR1 register must be cleared and respective interrupt control bits (ADIE in PIE1, GIE, PEIE in INTCON registers) may be enabled if required to generate interrupts.

After the completion of acquisition time the A/D conversion process must be started by setting the GO/DONE bit in ADCON0 (Bit 2 in PIC18458, Bit 1 in PIC18C801/ PIC18F8720) register. Since SAR type converts bits wise from MSB, a definite conversion time is required before the A/D conversion is complete. This time depends on the applied clock. Once the conversion is complete the converted result is stored in the result registers, GO/DONE bit in ADCON0 is reset/cleared and ADIF flag in PIR1 register is set automatically. The turn ON of A/D module and start of A/D conversion using GO/DONE should not be performed together in a single instruction. i.e. ADON and GO/DONE bits should not be set together in one instruction.

All the three PIC devices discussed have ADC resolution of 10-bit. A 16-bit register pair that is combination of two 8-bit registers A/D Result High (ADRESH) and A/D Result Low

(ADRESL) store the 10-bit converted result. The 10-bit result could be right or left justified, i.e. stored in the higher 10-bits or the lower 10-bits of the combined 16-bit ADRESH & ADRESL registers. The A/D Format Select (ADFM) bit (ADCON1<7> in PIC18F458 and ADCON2<7> in PIC18C801/PIC18F8720) controls the justification. When it is set to 1, the result is right justified; it is left justified when set to 0. This is shown in fig 12.7.

If polling is used instead of interrupts, the user's software must check and wait till the conversion is complete. i.e. wait till the GO/DONE bit in ADCON0 is cleared. The converted result may then be read and stored in RAM file registers through user's program. If interrupts are enabled the user software may proceed with its other tasks while the GO/DONE bit in ADCON0 is cleared and the read of result registers may be performed as part of ADC's ISR automatically. The results registers must be read before the completion of the next conversion. Otherwise, the earlier converted result is overwritten.



Fig 12.7 A/D Result Justification

A/D Acquisition & Conversion requirements:

An equivalent model of an analog input pin is shown in figure 12.8. The parasitic parameters block represents the resistance and capacitance seen at the input pin. The PIC microcontrollers include protection circuitry that protect from under or over voltages through clamped diodes at the analog input. The sample and hold circuit is represented by the switch SS and the capacitor C_{HOLD}. The switch isolates the holding capacitor from analog input pin.



Fig 12.8 A/D Analog Input Circuit

The C_{HOLD} capacitor is allowed to charge when the SS is closed and made to hold the value that is equivalent to analog input voltage. The time taken to charge the C_{HOLD} capacitor depends on the input impedances/parasitic parameters and need to be held at permissible levels. When the conversion is started, the sampling switch ss is opened and isolated from input pin so that the holding charge (voltage in C_{HOLD}) is held constant till the end of conversion. The switch is connected back after the conversion is complete, based on the selected clock. When the analog input channel is changed, a minimum acquisition time is required before a conversion can be started. The acquisition time is given by the expression assuming ½ LSB error with 1024 steps for 10-bit resolution.

T_{ACQ} = Amplifier Settling Time (T_{AMP}) +
Holding Capacitor Charging Time (T_C) +
Temperature Coefficient (T_{COEF}).

The minimum time values are as follows:

Amplifier Settling Time T_{AMP} around	2 µs
Holding capacitor charging time T_C around	9.61 µs
The temperature coefficient around	0.05 µs/^{0}C, and is neglected for T<25^{0}C
The acquisition time T_{ACQ} around	12.86 µs

The operation of A/D converter from the setting of GO bit to the clearing of DONE bit is shown in fig 12.9. The T_{AD} is defined as the time required to perform A/D conversion per bit. A total of 12 T_{AD} is required per 10-bit conversion. The actual $_{TAD}$ time varies based on selected clock as per the scaling of Table 12.5 and applied oscillator frequency. A minimum T_{AD} time of 1.6 µs is required for correct A/D conversions.

The minimum acquisition time is around 12.86 µs and minimum T_{AD} time is 1.6 µs. Hence any oscillation frequency resulting in TAD less than 1.6 µs should not be used as it violates minimum T_{AD} time, i.e. lower frequencies should be chosen. Similarly, any oscillation frequency resulting in TAD more than 12.86 µs is slow as conversion is taking more time than the acquired time. So, a higher frequency should be chosen. The same is demonstrated in table 12.9.



Fig 12.9 A/D Conversion T_{AD} cycles

Table 12.9 TAD vs Device Oscillator Frequencies

ADCS2: ADCS0	**Clock Conversion**	**Oscillator Frequency in MHz**					
		20	**10**	**5**	**2**	**1.25**	**1**
000	2 TOSC	100 ns(1)	200 ns(1)	400 ns(1)	1 µs(1)	1.6 µs	2 µs
100	4 TOSC	200 ns(1)	400 ns(1)	800 ns(1)	2 µs	3.2 µs	4 µs
001	8 TOSC	400 ns(1)	800 ns(1)	1.6 µs	4 µs	6.4 µs	8 µs
101	16 TOSC	800 ns(1)	1.6 µs	3.2 µs	8 µs	12.8 µs	16 µs(2)
010	32 TOSC	1.6 µs	3.2 µs	6.4 µs	16 µs(2)	25.6 µs(2)	32 µs(2)
110	64 TOSC	3.2 µs	6.4 µs	12.8 µs	32 µs(2)	51.2 µs(2)	64 µs(2)
011	RC	4 µs	4 µs	4 µs	4 µs	4 µs	4 µs

Note 1) These do not satisfy the minimum TAD requirement of 1 µs.
Lower frequencies or a high scaling needs to be used.
2) These are too slow as the T_{AD} time is more than T_{ACQ} time.
Higher frequencies or a low scaling needs to be used.
Also

Note:

- ADC module can operate in sleep mode and to operate the ADC in sleep mode, the ADC conversion clock must be derived from A/D's internal RC oscillator.
- Any clear of GO/$\overline{\text{DONE}}$ bit aborts the current conversion process and a 2 TAD cycles are required before a new acquisition after which acquisition starts automatically.
- When a port pin is in output mode and is used for A/D conversion as an analog input, the digital output level (VOH or VOL) will be converted.
- While reading Port registers such as PORTA, (PORTE or PORTF/PORTH), the pins configured as analog input are read as cleared (low level).
- Pins configured as digital input convert an analog input and may result in current consumption by the input buffer to be beyond the device specifications.

Special Event Trigger of CCP/ECCP

The CCP2 module of PIC18C801 & PIC18F8720, CCP1 or ECCP1 modules of PIC18F458 can trigger a special event using which an A/D conversion can be started, and the corresponding timers (Timer1 or Timer3) could be reset. This feature facilitates automatic A/D conversion for a precise sampling frequency with minimal software overhead. The special event trigger is configured when CCPxM3:CCPxM0 bits of CCPxCON<3:0> or ECCP1M3:ECCP1M0 bits of ECCP1CON<3:0> are set to ‘1011’ respectively. When a trigger occurs, the GO/DONE bit would be set to start the conversion process and the respective Timer1 (or Timer3) would be reset to zero. Also, the A/D module must be enabled, appropriate analog input channel is to be selected considering the minimum acquisition time and clock scaling for the special event trigger to work. In special event trigger mode, even when the A/D module is not enabled, the timer1 (or Timer3) will be reset, but the special event trigger for A/D conversion will be ignored.

Registers related to Capture, Compare, PWM, Timer1, Timer 2, Timer3 & Timer4

Name	Bit 7	Bit 6	Bit 5	Bit 4	Bit 3	Bit 2	Bit 1	Bit 0
ADRESH	ADRESH A/D Result Register High Byte							
ADRESL	ADRESL A/D Result Register Low Byte							
TRISA	—	PORTA Data Direction Register						
Only on PIC18F458								
ADCON0	ADCS1	ADCS0	CHS2	CHS1	CHS0	GO/ DONE	—	ADON
ADCON1	ADFM	ADCS2	—	—	PCFG3	PCFG2	PCFG1	PCFG0
TRISE	IBF	OBF	IBOV	PSP MODE	—	TRISE2	TRISE1	TRISE0
Only on PIC18C801 & PIC18F8720								
ADCON0	—	—	CHS3	CHS2	CHS1	CHS0	GO/ DONE	ADON
ADCON1	—	—	VCFG1	VCFG0	PCFG3	PCFG2	PCFG1	PCFG0
ADCON2	ADFM	—	—	—	—	ADCS2	ADCS1	ADCS0
TRISF	PORTF Data Direction Control Register							
TRISH	PORTH Data Direction Control Register							
All Devices								
INTCON	GIE/ GIEH	PEIE/ GIEL	TMR0IE	INT0IE	RBIE	TMR0IF	INT0IF	RBIF
PIR1	PSPIF(2)	ADIF	RCIF	TXIF	SSPIF	CCP1IF	TMR2IF	TMR1IF
PIE1	PSPIE(2)	ADIE	RCIE	TXIE	SSPIE	CCP1IE	TMR2IE	TMR1IE
IPR1	PSPIP(2)	ADIP	RCIP	TXIP	SSPIP	CCP1IP	TMR2IP	TMR1IP
PIR2	—	CMIF(2)	—	EEIF(2)	BCLIF	LVDIF	TMR3IF	CCP2IF(1)
PIE2	—	CMIE(2)	—	EEIE(2)	BCLIE	LVDIE	TMR3IE	CCP2IE(1)
IPR2	—	CMIP(2)	—	EEIP(2)	BCLIP	LVDIP	TMR3IP	CCP2IP(1)

Note 1: In PIC18F458 used for ECCP1, not CCP2. Note 2: Not present in PIC18C801

Configuring Setup steps for ADC Operation

1. Configure the analog input channel pins as inputs by setting the respective TRIS register bits.
2. Configure the analog pins and voltage reference and digital I/O pins using ADCON1 register.
3. Select the required A/D input channel for conversion using ADCON0 register.
4. Select the appropriate A/D conversion clock using ADCON0 (or ADCON2 in PIC18C801/PIC18F8720) register.
5. Enable/Turn ON the A/D module using ADCON0.
6. Clear the ADIF flag bit in PIR1<6> register.
7. If required set the control bits GIE, PEIE in INTCON<7:6> register and ADIE in PIE1<6> register to Enable the A/D interrupt.
8. Wait for the required acquisition time of 12.86 μs by including delay/NOP etc.
9. Start the A/D conversion by setting the GO/DONE bit in ADCON0 register.
10. If technique used is Polling - Wait for the ADIF flag to be set.

    (or)

    If interrupts are used, wait for the A/D interrupt.
11. Read the A/D Result registers (ADRESH & ADRESL) and store in RAM file register.
12. Clear the ADIF flag in PIR1<6> register.
13. Repeat from step 1,2 or 3 if next conversion is required. A minimum of $2T_{AD}$ is required between end of first and the start of the next conversion.

Program 12.5 Acquire sensor signal at analog input channel 0 at AN0, store the digital values in file registers 0x11, 0x10 and send the value to PORTC and PORTD.

```
        #INCLUDE<P18F458.INC>       ; Can be other MC
        ORG     0X00                ; Set Origin 0x0000 for Reset
        GOTO    START               ; Go to Start the Main program.
        ORG     0x30                ; Set Origin 0x0030 for Main Program Start
DELAY   MOWF    0x12,0              ; [0x12] = WREG
LM1:    NOP                         ; No Operation
        DECF    0X12, F,0           ; [0x12] = [0x12] - 1, Decrement file register
                                    ; 0x12 and store in the same file register.
        BNZ     LM1                 ; Branch to LM1 if Z flag is not set to 1.
        RETURN                      ; Return from Delay module
        ORG     0X50                ; Set Origin 0x0050 for Main Program
START   CLRF    TRISC               ; Make Port C as output, TRISC = 0x00
        CLRF    TRISD               ; Make Port D as output, TRISD = 0x00
        BSF     TRISA,0             ; Make Port A Pin 0 as input for Analog Input,
                                    ; TRISA<0> = 1
        MOVLW   0X8E                ; WREG = 0x8E
        MOVWF   ADCON1              ; ADCON1 = 0x8E, Right Justified, ADCS2 = 0 for
                                    ; selecting scaling 32, Make RA0/AN0 as analog
                                    ; input and other pins as Digital I/O
```

```
        MOVLW   0X81            ; WREG = 0x81, 20 MHz, 32 scaling → 1.6µs.
        MOVWF   ADCON0          ; ADCON0 = 0x81, ADCS<1:0> = 10 for selecting
                                ; scaling 32, Select Channel 0 AN0 as analog
                                ; input for A/D conversion, Enable A/D module
UP      BCF     PIR1, ADIF      ; PIR1<6> = 0, Clear ADIF flag in PIR1 register.
        MOVLW   0x08            ; For approx. 64 cycles, 12.88 µs at 20 MHz
        CALL    DELAY           ; Wait for Acquisition time. 12.86 µs. Delay
                                ; before start of conversion
        BSF     ADCON0, GO      ; Start A/D Conversion.
L1      BTFSC   ADCON0,DONE     ; Wait for A/D conversion to complete, Check
                                ; for Done bit to go Cleared/Low. Skip next
                                ; instruction if Done is cleared
        ; BTFSC PIR1, ADIF      ; Wait for A/D conversion to complete, Check
                                ; for ADIF flag to be Set/High, Skip next
                                ; instruction if event flag
        BRA     L1              ; Branch to L1 if flag not set/Done not cleared.
        MOVF    ADRESL, W       ; WREG = ADRESL, Store Low conv result
        MOVWF   0x10,0          ; [0x10] = WREG = ADRESL
        MOVF    ADRESH, W       ; WREG = ADRESH, Store High conv result
        MOVWF   0x11,0          ; [0x11] = WREG = ADRESH
        MOVFF   ADRESL, PORTC   ; PORTC = ADRESL
        MOVFF   ADRESH,PORTD    ; PORTD = ADRESH
        MOVLW   0x02            ; For approx. 16 cycles, 2 µs at 20 MHz
                                ; WREG = 0x02, inter conversion delay. 2 µs.
        CALL    DELAY           ; Wait for inter conversion delay. 2 µs.
        GOTO    UP              ; Goto UP, Repeat this process continuously.
        END
```

Program 12.6 Program of 12.5 with interrupts

```
        #INCLUDE<P18F458.INC>   ; Can be other MC
        ORG     0X00            ; Set Origin 0x0000 for Reset
        GOTO    START           ; Go to Start the Main program.
        ORG     0X08            ; Set Origin 0x0008 for Reset
        BTFSS   PIR1, ADIF      ; Wait for A/D conversion to complete,
                                ; Check for ADIF flag to be Set/High,
                                ; Skip next instruction if event flag
        RETFIE                  ; Return from Interrupt
        GOTO    ADC_ISR         ; Go to ADC_ISR
        ORG     0X30            ; Set Origin 0x0030 for Main Program
DELAY   MOWF    0x12,0          ; [0x12] = WREG
LM1:    NOP                     ; No Operation
        DECF    0X12, F,0       ; [0x12] = [0x12] - 1, Decrement file register
                                ; 0x12 and store in the same file register.
```

```
         BNZ      LM1             ; Branch to LM1 if Z flag is not set to 1.
         RETURN                   ; Return from Delay module
         ORG      0X50            ; Set Origin 0x0050 for Reset
ADC_ISR  MOVF     ADRESL, W       ; WREG = ADRESL, Store Low conv result
         MOVWF    0x10,0          ; [0x10] = WREG = ADRESL
         MOVF     ADRESH, W       ; WREG = ADRESH, Store High conv result
         MOVWF    0x11,0          ; [0x11] = WREG = ADRESH
         MOVFF    ADRESL, PORTC   ; PORTC = ADRESL
         MOVFF    ADRESH,         ; PORTD = ADRESH
                  PORTD
         MOVLW    0x02            ; For approx. 16 cycles, 2 µs at 20 MHz
                                  ; WREG = 0x02, inter conversion delay. 2 µs.
         CALL     DELAY           ; Wait for inter conversion delay. 2 µs.
         BCF      0x13,0          ; Clear Bit 0 in file register 0x13
         BCF      PIR1, ADIF      ; PIR1<6> = 0, Clear ADIF flag in PIR1 register
         RETFIE                   ; Return from Interrupt
         ORG      0x0200          ; Set Origin 0x0200 for Main Program Start
START    CLRF     TRISC           ; Make Port C as output, TRISC = 0x00
         CLRF     TRISD           ; Make Port D as output, TRISD = 0x00
         BSF      TRISA,0         ; Make Port A Pin 0 as I/P for Analog Input,
                                  ; TRISA<0> = 1
         MOVLW    0X8E            ; WREG = 0x8E
         MOVWF    ADCON1          ; ADCON1 = 0x8E, Right Justified, ADCS2 = 0
                                  ; for selecting scaling 32, Make RA0/AN0 as
                                  ; analog input and other pins as Digital I/O
         MOVLW    0X81            ; WREG = 0x81, 20 MHz, 32 scaling à 1.6µs.
         MOVWF    ADCON0          ; ADCON0 = 0x81, selecting ADCS<1:0> = 10
                                  ; for scaling 32, Channel 0 AN0 as analog I/P
                                  ; for A/D conversion, Enable A/D module
         BSF      INTCON, GIE     ; Global interrupt enable
         BSF      INTCON, PEIE    ; Peripheral interrupt enable
         BSF      PIR1, ADIE      ; ADC Interrupt enable
         BSF      IPR1, ADIP      ; Make ADC higher priority
         BCF      PIR1, ADIF      ; PIR1<6> = 0, Clear ADIF flag in PIR1 register
UP:      BTFSC    0x13,0          ; Check bit 0 of file register 0x13, if cleared
                                  ; Skip next instruction if cleared
         BRA      L2              ; Branch to L2, when bit 0 set in reg 0x13
         MOVLW    0x08            ; For approx. 64 cycles, 12.88 µs at 20 MHz
         CALL     DELAY           ; Wait for Acquisition time. 12.86 µs. Delay
                                  ; before start of conversion
         BSF      0x13,0          ; Set Bit 0 in file register 0x13
         BSF      ADCON0, GO      ; Start A/D Conversion.
L2       NOP                      ; Do optional other task
         GOTO     UP              ; Goto UP, Repeat this process continuously.
         END
```

Program 12.7 For PIC18C801/PIC18F8720 - Acquire 5 sensor signals at analog input channel 0 to 4 at AN0 to N4, send the converted value to PORTC and PORTD.

```
        #INCLUDE<P18F8720.INC>     ; Can be other MC
        ORG       0X00             ; Set Origin 0x0000 for Reset
        GOTO      START            ; Go to Start the Main program.
        ORG       0x30             ; Set Origin 0x0000 for Reset
DELAY   MOWF      0x12,0           ; [0x12] = WREG
LM1:    NOP                        ; No Operation
        DECF      0X12, F,0        ; [0x12] = [0x12] - 1, Decrement file register
                                   ; 0x12 and store in the same file register.
        BNZ       LM1              ; Branch to LM1 if Z flag is not set to 1.
        RETURN                     ; Return from Delay module
        ORG       0x0050           ; Set Origin 0x0050 for Main Program Start
START   CLRF      TRISC            ; Make Port C as output, TRISC = 0x00
        CLRF      TRISD            ; Make Port D as output, TRISD = 0x00
        SETF      TRISA            ; Make Port A as I/P for Analog I/P, TRISA = 1
        MOVLW     0x0A             ; WREG = 0x0A
        MOVWF     ADCON1           ; ADCON1 = 0x0A, Make RA0/AN0 to
                                   ; RA4/AN4
                                   ; as analog input and other pins as Digital I/O
                                   ; Select internal voltage reference AVDD & VSS
        MOVLW     0x00             ; WREG = 0x00
        MOVWF     ADCON0           ; ADCON0 = 0x00, Select RA0/AN0 channel
                                   ; for conversion
        MOVLW     0x82             ; WREG = 0x82
        MOVWF     ADCON2           ; ADCON2 = 0x82, ADCS<2:0> = 010, 20 MHz,
                                   ; 32 scaling → 1.6µs
        BSF       ADCON0,ADON      ; Enable ON A/D module
UP      BCF       PIR1, ADIF       ; PIR1<6> = 0, Clear ADIF flag in PIR1 register.
        MOVLW     0x08             ; For approx. 64 cycles, 12.88 µs at 20 MHz
        CALL      DELAY            ; Wait for Acquisition time. 12.86 µs. Delay
                                   ; before start of conversion
        BSF       ADCON0, GO       ; Start A/D Conversion.
L1      BTFSC     ADCON0,DONE      ; Wait for A/D conversion to complete, Check
                                   ; for Done bit to go Cleared/Low. Skip next
                                   ; instruction if Done is cleared
        BRA       L1               ; Branch to L1 if flag not set/Done not cleared.
        MOVFF     ADRESL, PORTC    ; PORTC = ADRESL
        MOVFF     ADRESH,PORTD     ; PORTD = ADRESH
        MOVLW     0x02             ; For approx. 16 cycles, 2 µs at 20 MHz
                                   ; WREG = 0x02, inter conversion delay. 2 µs.
```

```
CALL    DELAY         ; Wait for inter conversion delay. 2 µs.
MOVLW   0x04          ; WREG = 0x04
ADDWF   ADCON0,1,0    ; ADD 0x04 to ADCON0, i.e. keep on going to
                      ; selecting channel, i.e. ch = ch + 1
MOVLW   0x015         ; WREG = 0x15
SUBWF   ADCON0,0      ; Subtract, i.e. WREG = ADCON0-WREG
BNZ     UP            ; Branch to UP if not 0, i.e. till AN4 channel is
                      ; converted.
MOVLW   0x01          ; WREG = 0x01
ANDWF   ADCON0,0      ; Select channel 0 with A/D module ON
GOTO    UP            ; Goto UP, Repeat this process continuously.
END
```

12.3 Comparators & Configuration

Analog comparator is another mixed signal feature of PIC18 devices. The PIC18C801 does not contain an in-built comparator, whereas PIC18F458 and PIC18F8720 contains two analog comparators and a comparator voltage reference module.



a) Comparator



b) Sample Input and Output Waveform Non-Inverted



c) Sample Input and Output Waveform Inverted

Fig 12.10 Comparator symbol, Sample Input, Inverted & Non inverted Output Waveforms

Each analog comparator has two inputs VIN- and VIN+. The inputs may be applied through port pins. The voltage reference signal or ground may also be applied as one of the inputs. The comparator operating mode bits CM2:CM0 of CMCON <2:0> configures the input and outputs of comparators. The list of operating modes is presented in Table 12.10. The comparator control register format is presented next.

Table 12.10: Comparator Output Mode List

CM2:CM0	**Description** (Refer to Table & Fig for details)
000	Comparators Reset (POR Default Value)
001	One Independent Comparator with Output
010	Two Independent Comparators
011	Two Independent Comparators with Outputs
100	Two Common Reference Comparators
101	Two Common Reference Comparators with Outputs
110	Four Inputs Multiplexed to Two Comparators
111	Comparators Off

CMCON: Comparator Control Register Format

R-0	R-0	R/W-0	R/W-0	R/W-0	R/W-0	R/W-0	R/W-0
C2OUT	C1OUT	C2INV	C1INV	CIS	CM2	CM1	CM0
bit 7							bit 0

Bit No	**Bit**	**Description**
bit 7	C2OUT	**Comparator 2 Output bit** When C2INV = 0: 1 = C2 VIN+ > C2 VIN- 0 = C2 VIN+ < C2 VIN- When C2INV = 1: 1 = C2 VIN+ < C2 VIN- 0 = C2 VIN+ > C2 VIN-
bit 6	C1OUT	**Comparator 1 Output bit** When C1INV = 0: 1 = C1 VIN+ > C1 VIN- 0 = C1 VIN+ < C1 VIN- When C1INV = 1: 1 = C1 VIN+ < C1 VIN- 0 = C1 VIN+ > C1 VIN-
bit 5	C2INV	**Comparator 2 Output Inversion bit** 1 = C2 output inverted 0 = C2 output not inverted
bit 4	C1INV	**Comparator 1 Output Inversion bit** 1 = C1 output inverted 0 = C1 output not inverted

bit 3	CIS	**Comparator Input Switch bit** When CM2:CM0 = 110: 1 = C1 VIN- connects to RD0/PSP0 C2 VIN- connects to RD2/PSP2 0 = C1 VIN- connects to RD1/PSP1 C2 VIN- connects to RD3/PSP3
bit 2-0	CM2:CM0	**Comparator Mode bits.** Table 12.10 and Fig 12.8 shows the Comparator modes and CM2:CM0 bit settings

The status of Port Pins associated with Comparators in different modes is presented in Table 12.11. Here A indicates that the respective port pin is used as input to the analog comparator and D I/O indicates the port could be used for digital I/O. The output status of the two comparators is indicated through the C1OUT and C2OUT bits of CMCON <7:6>register. These bits are read only.

The output of the comparator could be configured to produce an inverting or non-inverting signal using the two control bits CMCON<4:5> register i.e. C1INV or C2INV for the respective comparators 1 and 2. Sample output waveforms are presented in fig 12.10 b and 12.10 c and the output levels for different cases has been presented along with comparator control register CMCON <6:7> format. When VIN+ > VIN- the output is high in non-inverting mode, and it is low otherwise. When VIN+ > VIN- the output is low in inverting mode, and it is high otherwise. The input offsets, drift and response may vary at the comparator, and this is shown as multi-line during the edges of waveform in fig 12.10 b and c.

Table 12.11: Status of Port Pins associated with Comparators in different modes.

Comparator		**Comparator 1**			**Comparator 2**		
PIN *		**PIN 1**	**PIN 2**	**PIN 3**	**PIN 4**	**PIN 5**	**PIN 6**
PIC18F458 Port Pins		RD1	RD0	RE1	RD3	RD2	RE2
PIC18F8720 Port Pins		RF6	RF5	RF2	RF4	RF3	RF1
PORT Pin Status for different Comparator I/O Operating Modes	000	A	A	D I/O	A	A	D I/O
	001	A	A	C1OUT	D I/O	D I/O	D I/O
	010	A	A	D I/O	A	A	D I/O
	011	A	A	C1OUT	A	A	C2OUT
	100	A	A	D I/O	A	D I/O	D I/O
	101	A	A	C1OUT	A	D I/O	C2OUT
	110	A	A	D I/O	A	A	D I/O
	111	D I/O	D I/O	D I/O	D I/O	D I/O	D I/O

Note *: Here PIN 1 to PIN 6 are not the respective pin numbers on the IC package.

Table 12.12: Comparator Input/Output Mode Pin selection.

Comparator Input Terminal		**Comparator 1**			**Comparator 2**		
		VIN-	**VIN+**	**C1OUT**	**VIN-**	**VIN+**	**C2OUT**
	000	PIN 1	PIN 2	OFF (1)	PIN 4	PIN 5	OFF (1)

Comparator I/O Operating Modes	001	PIN 1	PIN 2	C1OUT/ PIN 3 (2)	GND	GND	OFF (1)
	010	PIN 1	PIN 2	C1OUT	PIN 4	PIN 5	C2OUT
	011	PIN 1	PIN 2	C1OUT/ PIN 3 (2)	PIN 4	PIN 5	C2OUT/ PIN 6 (2)
	100	PIN 1	PIN 2	C1OUT	PIN 4	PIN 2	C2OUT
	101	PIN 1	PIN 2	C1OUT/ PIN 3 (2)	PIN 4	PIN 2	C2OUT/ PIN 6 (2)
	110	PIN 1 OR PIN 2	CVREF	C1OUT	PIN 4 OR PIN 5	CVREF	C2OUT
	111	GND	GND	OFF (1)	GND	GND	OFF (1)

(1) Indicates comparator output is disabled/OFF, Read as 0

(2) Indicates the comparator output is also routed to external port pin

The coloured cells indicate deviation from normal.



Fig 12.11 Comparator Block Diagram

The Input/Output and mode selection of comparators is presented in Table 12.12 and connections is shown in fig 12.11. In mode 0 two comparators are in reset mode, and all four inputs are connected to PIN 1 to PIN 4. In mode 7 the four inputs are connected to GND. In mode 0 or mode 7 comparators outputs are OFF.



a) CM2:CM0 = 000
Comparators Reset



b) CM2:CM0 = 001
One Independent Comparator with Output



c) CM2:CM0 = 010
Two Independent Comparators



d) CM2:CM0 = 011
Two Independent Comp with Outputs



e) CM2:CM0 = 100
Two Common Reference Comparators



f) CM2:CM0 = 101
Two Common Reference Comp with O/Ps



g) CM2:CM0 = 110
Four Inputs Multiplexed to Two Comps



h) CM2:CM0 = 111
Comparators Off

Fig 12.12 Comparator I/O Operating Modes

In both these modes the comparator outputs C1OUT and C2OUT are disabled/OFF and are read as 0. In other modes the C1OUT (mode 1 to 6) and C2OUT (mode 2 to 6) are active. In modes 2, 4, 6 both the two comparators are active and the outputs C1OUT and C2OUT are not routed to PIN 3 or PIN 6. Hence the PIN 3 and PIN 6 could be used as digital I/O's in mode 0, 2, 4, 6, 7.

In mode 1 only comparator 1 is active and comparator 2 is OFF. Since comparator 2 is inactive, the PIN 4, 5, 6 could be used as digital I/O in mode 1. In mode 3, 5 both the comparators are active. The comparator 1 output C1OUT is asynchronously routed to PIN 3 in mode 1, 3, 5 and comparator 2 output C2OUT is also routed to PIN 6 in mode 3 and 5.

Additionally, in mode 4 and mode 5 the PIN 5 could be used as digital I/O since the input VIN+ for the comparator 2 is also taken from PIN 2. In mode 6, the VIN+ is tied to comparator voltage reference CVREF generated by the comparator voltage reference module. The VIN- input of comparator 1 could be selected from PIN 1 or PIN 2 and from PIN 4 or PIN 5 for comparator 2. The CIS bit in CMCON <3> register is used for this selection. The CVREF generated by comparator voltage reference module makes uses of internal VDD & VSS or external VREF+ & VREF- provided through AN3/RA3 & AN2/RA2 port pins. The details of CVREF generation are presented in comparator voltage reference module in the next section.

The comparators input and output pins in each of the 8 modes must be configured using the TRIS data direction control register. A separate interrupt flag or control is not available for each of the comparator 1 and 2. Both the comparators 1 and 2 are provided with the same CMIF flag and control bits. A change in any one of the comparator outputs sets the CMIF flag in PIR2 register and generates an interrupt if enabled (GIE, PEIE & CMIE).

The User's software must determine the comparators source (comparator 1 or comparator 2) that caused the interrupt by checking the C1OUT and C2OUT bits of CMCON<6:7> register. The CMIF flag is set irrespective of whether the interrupt is enabled or not. The user may clear the interrupt in ISR either reading or writing of CMCON (to end the mismatch condition) or by clearing the CMIF flag bit. An interrupt from comparator can wake-up the device from sleep mode if enabled. Wake-up from sleep mode does not affect the contents of CMCON register.

Registers related to Comparator1 and Comparator 2

Name	Bit 7	Bit 6	Bit 5	Bit 4	Bit 3	Bit 2	Bit 1	Bit 0
In PIC18F458 & PIC18F8720; (PIC18C801 doesn't have comparators)								
CMCON	C2OUT	C1OUT	C2INV	C1INV	CIS	CM2	CM1	CM0
CVRCON	CVREN	CVROE	CVRR	CVRSS	CVR3	CVR2	CVR1	CVR0
INTCON	GIE/ GIEH	PEIE/ GIEL	TMR0IE	INT0IE	RBIE	TMR0IF	INT0IF	RBIF

PIR2	—	CMIF(1)	—	EEIF(1)	BCLIF	LVDIF	TMR3IF	CCP2IF(2)
PIE2	—	CMIE(1)	—	EEIE(1)	BCLIE	LVDIE	TMR3IE	CCP2IE(2)
IPR2	—	CMIP(1)	—	EEIP(1)	BCLIP	LVDIP	TMR3IP	CCP2IP(2)
In PIC18F458								
PORTD	RD7	RD6	RD5	RD4	RD3	RD2	RD1	RD0
TRISD	PORTD Data Direction Register							
PORTE	—	—	—	—	—	RE2	RE1	RE0
TRISE	IBF	OBF	IBOV	PSPMODE	—	TRISE2	TRISE1	TRISE0
In PIC18F8720								
PORTF	RF7	RF6	RF5	RF4	RF3	RF2	RF1	RF0
TRISF	PORTF Data Direction Register							

Note 1: Not present in PIC18C601/801.

Note 2: CCP2 not in PIC18F458, Used for ECCP1.

Note:

- Change of comparator mode may result in a false interrupt if interrupts are enabled. Hence the comparator interrupt must be disabled before a mode is changed.
- Also, a minimum time delay is to be allowed before an effect of change in mode could be observed.
- A device reset causes the comparator resets the CMCON and makes the comparator to be in reset mode – CM2:CM0 with value of 000 and configures all the comparator input pins as analog input pins so as to reduce device current.
- A pin that is configured as digital input, when subjected to analog levels results in higher current consumption by the input buffer.
- If a port register is read when some of the pins are configured as analog inputs, they are read as 0.
- When a comparator is enabled while the device is in sleep mode, higher sleep current is consumed beyond the power-down current specifications.
- To reduce current consumption, Comparators need to be turned off by setting CM2:CM0 to 111.

Program 12.8 For PIC18F458 compare an analog signal at VIN+ of comparator1 to determine the rise time i.e. 0%-90% where an analog equivalent of 90% VDD is applied at VIN- of the comparator1. The time taken in terms of number of timer pulses is sent to PORTB and PORTC. The output of the comparator RE1/C1OUT may serve as an enable signal to apply new signal for measuring rise time again as shown in fig P12.8.



Fig P12.8 Comparator connection to determine Rise time 0-90% ideal.

```
        ORG     0x0000          ; Set Origin 0x0000 of Flash memory
        GOTO    START           ; Go to Start of Main program.
        ORG     0x0030          ; Set Origin 0x0030 for Main program
START   CLRF    TRISB           ; Make Port B as output
        CLRF    TRISC           ; Make Port C as output
        SETF    TRISD           ; Make Port D as input (Comparator inputs)
        CLRF    TRISE           ; Make Port E as output (Comparator outputs)
        MOVLW   0x01            ; WREG = 0x01

        MOVWF   CMCON           ; Comparator 1 ON with O/P,
                                ; Comparator 2 OFF
        MOVLW   0x00            ; WREG = 0x00
        MOVWF   T1CON           ; T1CON =0x00, select 8-bit read/write
                                ; operations of Timer1, with Prescale 1,
                                ; select internal main clock fosc/4
UP      CLRF    TMR1H           ; Initialize the Timer 1 high count register
                                ; TMR1H to 0x00
        CLRF    TMR1L           ; Initialize the Timer 1 low count register
                                ; TMR1L to 0x00
        BCF     PIR2,CMIF       ; Clear CMIF Comparator flag
        BSF     T1CON,0         ; TMR1ON, Start Timer 1 for number of pulses
L1      BTFSS   PIR2,CMIF       ; Wait for crossover to have change in
                                ; comparator output, i.e. CMIF flag.
                                ; Skip next instruction if change occurs
        BRA     L1              ; Branch to L1 if flag not set
        BCF     T1CON,0         ;TMR1OFF, Stop the Timer 1.
        MOVFF   TMR1L, PORTC    ; Send TMR1L count to PORTC, first low is to
                                ; be read to make TMR1H available.
        MOVFF   TMR1H, PORTB    ; Send TMR1H count to PORTB
        GOTO    UP              ; Goto UP, Repeat this process; continuously.
        END
```

Program 12.9 Program of 12.8 with interrupts

```
        ORG     0x0            ; Set Origin 0x0 of Flash memory
        GOTO    START          ; Go to Start the Main program.
        ORG     0x08           ; Set Origin 0x08 High Priorty interrupt Vector
        BTFSS   PIR2,CMIF      ; Wait for crossover to have change in
                               ; comparator output, i.e. CMIF flag.
                               ; Skip next instruction if change occurs
        BRA     IL1            ; If not due to CMIF go to IL1, do nothing, ;
                               come out of ISR.
        GOTO    CM_ISR         ; Go to CM_ISR
IL1     NOP                    ; NO operation
        RETFIE                 ; Return from ISR, Go back to main program
        ORG     0x100          ; Set Origin 0x100 for CM_ISR
CM_ISR  BCF     T1CON,0        ;TMR1OFF ; Stop the Timer 1.
        MOVFF   TMR1L, PORTC   ; Send TMR1L count to PORTC, First low is to
                               ; be read to make TMR1H available.
        MOVFF   TMR1H, PORTB   ; Send TMR1H count to PORTB
        CLRF    TMR1H          ; Initialize the Timer 1 high count register
                               ; TMR1H to 0x00
        CLRF    TMR1L          ; Initialize the Timer 1 low count register
                               ; TMR1L to 0x00
        CLR     PIR2,CMIF      ; Clear CMIF Comparator flag
        BSF     T1CON,0        ; TMR1ON, Start Timer 1 for number of pulses
        RETFIE                 ; Return from ISR, Go back to main program
        ORG     0x200          ; Set Origin 0x200 for Main program
START   CLRF    TRISB          ; Make Port B as output
        CLRF    TRISC          ; Make Port C as output
        SETF    TRISD          ; Make Port D as input
        CLRF    TRISE          ; Make Port E as output
        MOVLW   0x01           ; WREG = 0x01
        MOVWF   CMCON          ; Comparator 1 ON with O/P,
                               ; Comparator 2 OFF
        MOVLW   0x00           ; WREG = 0x00
        MOVWF   T1CON          ; T1CON =0x00, select 8-bit read/write
                               ; operations of Timer1, with Prescale 1,
                               ; select internal main clock fosc/4
        BCF     PIR2,CMIF      ; Clear CMIF Comparator flag
        BSF     INTCON,GIE     ; Enable Global interrupt
        BSF     INTCON,PEIE    ; Enable Peripheral interrupts
        BSF     RCON,IPEN      ; Enable interrupt priorities
```

	BSF	IPR2,CMIP	; Make CMIF as high priority
	BSF	PIE2,CMIE	; Enable CMIF interrupt
	BSF	T1CON,0	; TMR1ON, Start Timer 1 for number of pulses
UP	NOP		; No operation, Logic of optional alternate ; function may be placed here.
	GOTO	UP	; Goto UP, Repeat this process; continuously.
	END		



Fig P12.10 Comparator connection to determine Rise time 10%-90% ideal.

Similarly fall time may found with VIN- held at 10% VDD. To determine 10% to 90% rise time two comparators may be used by configuring CM2:CM0 in mode 5 with the required signal applied at VIN+ of comparator 1 and 2 and VIN- of comparator 1 at 10% VDD and VIN- of comparator 2 at 90% VDD. Timer may be started after comparator 1 output changes at 10% VDD and stopped when comparator 2 output changes at 90% VDD.

Program 12.10. Determine 10% to 90% rise time of a signal.

	ORG	0x0000	; Set Origin 0x0000 of Flash memory
	GOTO	START	; Go to Start of Main program.
	ORG	0x0030	; Set Origin 0x0030 for Main program
START	CLRF	TRISB	; Make Port B as output
	CLRF	TRISC	; Make Port C as output
	SETF	TRISD	; Make Port D as input
	CLRF	TRISE	; Make Port E as output
	MOVLW	0x05	; WREG = 0x05
	MOVWF	CMCON	; Common referenced Comparator 1 and 2 ON ; with output
	MOVLW	0x00	; WREG = 0x00
	MOVWF	T1CON	; T1CON =0x00, select 8-bit read/write ; operations of Timer1, with Prescale 1, ; select internal main clock fosc/4
UP	CLRF	TMR1H	; Initialize the Timer 1 high count register ; TMR1H to 0x00

```
        CLRF    TMR1L          ; Initialize the Timer 1 low count register
                               ; TMR1L to 0x00
        BCF     PIR2,CMIF      ; Clear CMIF Comparator flag
                               ; Wait for comparator 1 crossover 10% to have
L1      BTFSS   PIR2,CMIF      ; change in comparator output, i.e. CMIF flag.
                               ; Skip next instruction if change occurs
        BRA     L1             ; Branch to L1 if flag not set
        BSF     T1CON,0        ; TMR1ON, Start Timer 1 for number of pulses
        BCF     PIR2,CMIF      ; Clear CMIF Comparator flag
                               ; Wait for comparator 2 crossover 90% to have
L2      BTFSS   PIR2,CMIF      ; change in comparator output, i.e. CMIF flag.
                               ; Skip next instruction if change occurs
        BRA     L2             ; Branch to L2 if flag not set
        BCF     T1CON,0        ;TMR1OFF, Stop the Timer 1.
        MOVFF   TMR1L, PORTC   ; Send TMR1L count to PORTC, first low is to
                               ; be read to make TMR1H available.
        MOVFF   TMR1H, PORTB   ; Send TMR1H count to PORTB
        GOTO    UP             ; Goto UP, Repeat this process; continuously.
        END
```

Also, the comparator output may be connected to CCP input for capture or A/D conversion.

12.4 DAC & Interfacing

Digital to analog or D/A Converter is another mixed signal feature that is included with microcontrollers. The signals that are given to actuators and control devices typically are analog in nature and the digital values (1's & 0's) produced microprocessors/ microcontrollers must be converted to analog signals before they are applied to the actuators or any other analog devices. A general interfacing is presented in Fig 12.13a.



Fig 12.13a) General Interfacing of actuators using DAC



Fig 12.13b) Internal block diagram of DAC 0808 (Source: TI DAC 0808)



Fig 12.14 Interfacing D/A converter to produce 10V Output

The D/A converters take a set of digital values (1's & 0's) and convert to an equivalent analog signal. The three PIC18 devices discussed here (PIC18F458, PIC18C801 &

PIC18F8720) does not have an in-built DAC but PIC18xxK40 & PIC18xxK42 devices have a built-in DAC. The details of the in-built DAC of K40/K42 are presented later.

The internal block diagram of DAC0808 is presented in fig 12.13b. concept of DAC's using external DAC0808 which makes use of R-2R ladder type DAC. A typical interfacing of the DAC0808 with a Microprocessor/PPI/Microcontroller is shown in fig 12.14.

The expression of D/A conversion for output current of given circuit is given as

$$I_O = K\left(\frac{A1}{2} + \frac{A2}{4} + \frac{A3}{8} + \frac{A4}{16} + \frac{A5}{32} + \frac{A6}{64} + \frac{A7}{128} + \frac{A8}{256}\right) \quad \text{Eq. 12.3.1}$$

$$\text{where } K \cong \frac{V_{REF}}{R} \quad \text{Eq. 12.3.2}$$

A1 represents MSB and A8 represents LSB of the 8-bit digital input and V_{REF} is the reference voltage typically 10V and R being the resistance typically 5KΩ. The expression of D/A conversion for output voltage of given circuit is given as

$$V_O = V_{REF}\left(\frac{A1}{2} + \frac{A2}{4} + \frac{A3}{8} + \frac{A4}{16} + \frac{A5}{32} + \frac{A6}{64} + \frac{A7}{128} + \frac{A8}{256}\right) \quad \text{Eq. 12.3.3}$$

Substituting V_{REF} = 10V

$$V_O = 10V\left(\frac{A1}{2} + \frac{A2}{4} + \ldots \frac{A8}{256}\right) \quad \text{Eq. 12.3.4}$$

Factor	Value	Factor	Value	Factor	Value	Factor	Value
1/2	0.5	1/8	0.125	1/32	0.03125	1/128	0.007813
1/4	0.25	1/16	0.0625	1/64	0.015625	1/256	0.003906

Using the corresponding factor & equivalent values, the full-scale output voltage is given as V_o = 10V * (.996) = 9.96V

Program 12.11. Generate a square wave having 0 to 5V with duty cycle of 60% when DAC is connected to PORT C. Waveform shown in Fig P12.11.

```
        ORG     0x0000  ; Set Origin 0x0000 of Flash memory
        GOTO    START   ; Go to Start of Main program.
        ORG     0x0030  ; Set Origin 0x0030 for Main program
START   CLRF    TRISC   ; Make Port C as output
UP      MOVLW   0x80    ; WREG = 0x80
        MOVWF   PORTC   ; PORTC = 0x80, Voltage level equivalent to 5V.
        MOVLW   0x60    ; WREG = 0x60, For 60% ON Time
        CALL    DELAY   ; Set Delay
        MOVLW   0x00    ; WREG = 0x00
        MOVWF   PORTC   ; PORTC = 0x00, Voltage level equivalent to 0V.
        MOVLW   0x40    ; WREG = 0x40, For 40% OFF Time
        CALL    DELAY   ; Set Delay
```

```
         GOTO     UP        ; Goto UP, Repeat this process; continuously.
DELAY    NOP                ; No Operation
L1       DECF     WREG      ; WREG = WREG-1
         BNZ      L1        ; if WREG! = 0 Branch to L1
         RET                ; Return from Delay module
         END
```

The interfacing circuit shown in fig 12.14 produces a 9.96V full scale output waveform when the 8-bit value is 0xFF. Half of that gives a close 5V and hence the 0x80 to represent a 5V level. Other voltage levels may be obtained by providing the appropriate ratio value to PORC register. Ex: 0x80 for a 5V or 50% of FSO, 0x40 for 2.5V or 25% of FSO, 0xC0 for 7.5V or 75% etc. The appropriate frequency may be achieved by setting the WREG value used for the respective ON time and OFF time before the calling of delay module. Time based calculations are presented in detail in the timer's section.



Fig P12.11 Square Waveform for P12.11 Duty cycle of 60% with 0 to 5V.

Program 12.12. Generate a triangular wave having 2.5V to 7.5V with DAC driven by PORTC. Waveform shown in Fig P12.12.

```
         ORG      0x0000    ; Set Origin 0x0000 of Flash memory
         GOTO     START     ; Go to Start of Main program.
         ORG      0x0030    ; Set Origin 0x0030 for Main program
START    CLRF     TRISC     ; Make Port C as output
         MOVLW    0x40      ; WREG = 0x40, Voltage level equivalent to 2.5V.
         MOVWF    0x10      ; File Reg 0x10 = 0x40
         MOVLW    0xC0      ; WREG = 0xC0, Voltage level equivalent to 7.5V.
         MOVWF    0x11      ; File Reg 0x11 = 0xC0
         MOVLW    0x40      ; WREG = 0x40, Starting Triangular at 2.5V.
L1       MOVWF    PORTC     ; PORTC = 0x40, Voltage level equivalent to 2.5V.
         INCF     WREG      ; WREG = WREG + 1
         CPFSEQ   0x11,0    ; Compare WREG == 0x11,
                            ; if equal skip next instruction
         BRA      L1        ; If not equal branch to L1
L2       MOVWF    PORTC     ; PORTC = 0xC0, Voltage level equivalent to 7.5V.
         DECF     WREG      ; WREG = WREG - 1
         CPFSEQ   0x10,0    ; Compare WREG == 0x10,
                            ; if equal skip next instruction
```

```
        BRA     L2      ; If not equal branch to L2
        GOTO    L1      ; Goto L1, Repeat this process; continuously.
        END
```



Fig P12.12 Triangular Waveform for P12.12 0 to 5V.

A particular frequency or slope may be pre-setting number of increment/decrement steps. This may be achieved by adding a finite step value more than 1 and considering the clock frequency and machine cycle. Time based calculations are presented in detail in the timer's section.



Fig P12.13 Triangular Waveform for P12.13 0 to 5V.

Program 12.13. Generate a staircase waveform with 5 steps of 0V, 1.25V, 2.5V, 5V, 7.5V when DAC is connected to PORT C. Waveform shown in Fig P12.13.

```
        ORG     0x0000  ; Set Origin 0x0000 of Flash memory
        GOTO    START   ; Go to Start of Main program.
        ORG     0x0030  ; Set Origin 0x0030 for Main program
START   CLRF    TRISC   ; Make Port C as output
 UP     MOVLW   0x00    ; WREG = 0x00, Voltage level equivalent to 0V.
        MOVWF   PORTC   ; PORTC = 0x00, Voltage level equivalent to 0V.
        MOVLW   0x20    ; WREG = 0x20, Voltage level equivalent to 1.25V.
        MOVWF   PORTC   ; PORTC = 0x20, Voltage level equivalent to 1.25V.
        MOVLW   0x40    ; WREG = 0x40, Voltage level equivalent to 2.5V.
        MOVWF   PORTC   ; PORTC = 0x40, Voltage level equivalent to 2.5V.
        MOVLW   0x80    ; WREG = 0x80, Voltage level equivalent to 5V.
        MOVWF   PORTC   ; PORTC = 0x80, Voltage level equivalent to 5V.
```

```
MOVLW   0xC0    ; WREG = 0xC0, Voltage level equivalent to 7.5V.
MOVWF   PORTC   ; PORTC = 0xC0, Voltage level equivalent to 7.5V.
GOTO    UP      ; Goto UP, Repeat this process; continuously.
END
```

A delay with an appropriate value may be used after every step to control the frequency of the steps in staircase.

12.5 PIC18's 5-Bit DAC & Configuration.

The PIC18xxK40/PIC18xxK42 devices have one in-built 5-Bit DAC. The PIC18xK40/PIC18xK42 devices have two variations.

- Regular Flash Variant
- Low Power Flash Variant

The regular Flash variant takes the form PIC18FxK40/K42 whereas low power flash variant takes the form PIC18LFxK40/K42, where x determines the internal member of the family which varies based on memory and other features.

The regular flash variant PIC18FxK40/K42 is operable between 2.3V to 5.5V for oscillator frequencies less than 16MHz, 2.5V to 5.5V for frequencies between 16MHz to 32MHz and 2.7V to 5.5V for frequencies greater than 32MHz. The low power flash variant PIC18LFxK40/K42 is operable between 1.8V to 3.6V for oscillator frequencies less than 16MHz, 2.5V to 3.6V for frequencies between 16MHz to 32MHz and 3V to 3.6V for frequencies greater than 32MHz. Hence the 3V is the typical voltage V_{DD}.

The DACs are configured and controlled by using the two control registers DAC1CON0 & DAC1CON1**.** The format of these DAC modules control registers is presented here.

DAC1CON0: DAC Control Register Format

R/W-0/0	U-0	R/W-0/0	R/W-0/0	R/W-0/0	R/W-0/0	U-0	R/W-0/0
EN	—	OE1	OE2	PSS<1:0>		—	NSS
bit 7							bit 0

Bit No	Bit	Description
bit 7	EN	DAC Enable bit 1 = DAC is enabled 0 = DAC is disabled
bit 6	Unimplemented	Read as '0'
bit 5	OE1	DAC Voltage Output Enable bit 1 = DAC voltage level is output on the DAC1OUT1 pin 0 = DAC voltage level is disconnected from DAC1OUT1 pin
bit 4	OE2	DAC Voltage Output Enable bit 1 = DAC voltage level is output on the DAC1OUT2 pin

		0 = DAC voltage level is disconnected from DAC1OUT2 pin
bit 3-2	PSS<1:0>	DAC Positive Source Select bit 11 = Reserved 10 = FVR buffer 01 = VREF+ 00 = AVDD
bit 1	Unimplemented	Read as ‘0’
bit 0	NSS	DAC Negative Source Select bit 1 = VREF- 0 = AVSS

DAC1CON1: DAC Data Register Format

U-0	U-0	U-0	R/W-0/0	R/W-0/0	R/W-0/0	R/W-0/0	R/W-0/0
—	—	—	DATA[4:0]				
bit 7							bit 0

Bit No	Bit	Description
bit 7-5	Unimplemented	Read as ‘0’
bit 4-0	DAC1R<4:0>	Data Input Register for DAC bits

The block diagram of in-built DAC of PIC18xxK40/PIC18xxK42 is shown in fig 12.15. The DAC’s makes uses of two input source connections.

- VSOURCE+ → Positive input Source
- VSOURCE- → Negative input Source

The DAC’s VSOURCE+ may be sourced from 3 different power sources. They are

- VDD supply voltage
- External VREF+ /AN3 pin
- FVR Buffer

  The FVR - Fixed Voltage Reference, is a stable voltage reference, independent of VDD. It has selectable output levels of 1.024V, 2.048V or 4.096V. The output level of FVR may be selected and used as supply reference voltage for the following:

  - ADC input channel
  - ADC positive reference
  - Comparator input
  - Digital-to-Analog Converter (DAC)

  The FVRCON register is used to configure the FVR module.

  Note: The FVR feature is similar to that of CVREF. The FVR is present in PIC18xxK40/PIC18xxK42 but is not present in the 3 PIC18 devices (PIC18F458, PIC18C801 & PIC18F8720) that are discussed in detail. An equivalent but different configurable Voltage Reference module is present in PIC18F458 & PIC18F8720.



Fig 12.15 Block diagram of an in-built DAC of PIC18xxK40/PIC18xxK42

The DAC's VSOURCE- may be sourced from 2 different power sources. They are

- External VREF-/AN2 pin
- Vss

The DAC Positive Source Select bits PSS<1:0> of DAC1CON0<3:2> register is used to select the DAC's positive source VSOURCE+. The DAC Negative Source Select bit NSS of DAC1CON0<0> is used to select the DAC's negative source VSOURCE-. The DAC is enabled through the EN bit of DAC1CON0<7> register.

The output of DAC (DACx_output) may be used as a reference source for the following:

- Comparator positive input
- ADC input channel
- DAC1OUT1/RA2 pin
- DAC1OUT2/RB7 pin

Note that the DAC1OUT1 and DAC1OUT2 outputs are not internally buffered and must be buffered externally through ex: a voltage follower. i.e. the DAC1OUT1/DAC1OUT2 pins drive the voltage follower, and the actuators are driven by the voltage follower as shown in Fig 12.16. The voltage follower here represents the signal conditioning circuit.

When the DAC is enabled by setting the bit EN to 1, then DAC output Voltage is given by

$$DACx\_output = \left((\mathrm{VREF+} - \mathrm{VREF-}) \times \frac{DACR[4:0]}{2^5} \right) + \mathrm{VREF-}$$

Eq. 12.3.5 [Source: K42]

For a typical reference voltages VREF+ of +3V and VREF- of GND, the DAC output is as follows in table 12.11



Fig 12.16 Block diagram of an in-built DAC of PIC18xK40/PIC18xK42

Table 12.13: Generated DAC analog output Voltage

Steps	Digital Input	Analog Voltage for respective Digital Input	
		VREF+ = VDD = 3V & VSS = VREF- = GND	VREF+ = VDD = 5V & VSS = VREF- = GND
0	0 0000	0 V	0 V
1	0 0001	0.09375 V	0.15625 V
2	0 0010	0.1875 V	0.3125 V
3	0 0011	0.28125 V	0.46875 V
4	0 0100	0.375 V	0.625 V
7	0 0111	0.65625 V	1.09375 V
8	0 1000	0.75 V	1.25 V
15	0 1111	1.40625 V	2.34375 V
16	1 0000	1.5 V	2.5 V
31	1 1111	2.90625 V	4.84375 V

Registers related to DAC

Name	Bit 7	Bit 6	Bit 5	Bit 4	Bit 3	Bit 2	Bit 1	Bit 0
DAC1CON0	EN	—	OE1	OE2	PSS<1:0>		—	NSS
DAC1CON1	—	—	—	DATA<4:0>				
FVRCON	FVREN	FVRRDY	TSEN	TSRNG	CDAFVR<1:0>		ADFVR<1:0>	

The register related to DAC is presented above. The FVRCON register is used to configure the FVR – Fixed voltage reference and DAC1CON0 and DAC1CON1 are used for configuring the DAC.

Program 12.14. Generate a square wave having 0 to 1.5V with duty cycle of 60% using in-built DAC of PIC18xxK40/PIC18xxK42 as shown in fig P14.14.

```
        #INCLUDE<PIC18F57K42.INC>  ; Can be other MC
        ORG     0x0000             ; Set Origin 0x0000 of Flash memory
        GOTO    START              ; Go to Start of Main program.
        ORG     0x0030             ; Set Origin 0x0030 for Main program
START   BCF     TRISA,2            ; Make Port A Pin 2 as output DAC1OUT1
        CLRF    TRISC              ; Make Port C as output
        MOVLW   0x80               ; WREG = 0x80
        MOVWF   DAC1CON0           ; Enable the DAC with AVDD and AVSS as
                                   ; source 1 and 2
        BSF     DAC1CON0,OE1       ; DAC Voltage level connected to DAC1OUT1
UP      MOVLW   0x10H              ; WREG = 0x10
        MOVWF   DAC1CON1           ; DAC1CON1 = 0x10 to get 1.5V.
        MOVLW   0x60               ; WREG = 0x60, For 60% ON Time
        CALL    DELAY              ; Set Delay
        MOVLW   0x00               ; WREG = 0x00
        MOVWF   DAC1CON1           ; DAC1CON1 = 0x10 to get 0V.
        MOVLW   0x40               ; WREG = 0x40, For 40% OFF Time
        CALL    DELAY              ; Set Delay
        GOTO    UP                 ; Goto UP, Repeat this process; continuously.
DELAY   NOP                        ; No Operation
L1      DECF    WREG               ; WREG = WREG-1
        BNZ     L1                 ; if WREG != 0 Branch to L1
        RET                        ; Return from Delay module
        END
```



Fig P12.14 Square Waveform for P12.14 Duty cycle of 60% with 0 to 5V.